システムプログラミング 1 期末レポート

氏名: 重近 大智 (SHIGECHIKA, Daichi) 学生番号: 09501527

> 出題日: 2020 年 10 月 05 日 提出日: 2020 年 11 月 14 日 締切日: 2020 年 11 月 16 日

1 概要

本レポートでは、MIPS 言語を用いて、提示された 5 つの課題に取り組み、その解答を報告する。実行結果は xspim による結果である。

本レポートで報告するシステムプログラミング1の課題は次の5つである.

- 1. A.8 節 「入力と出力」に示されている方法と,A.9 節 最後「システムコール」に示されている方法のそれ ぞれで "Hello World" を表示せよ.両者の方式を比較し考察せよ.[1]
- 2. アセンブリ言語中で使用する .data, .text および .align とは何か解説せよ. 6.2 節のコード中の 9 行目の .data がない場合, どうなるかについて考察せよ.
- 3. A.6 節 「手続き呼出し規約」に従って,再帰関数 fact を実装せよ. (以降の課題においては,この規約に 全て従うこと)[1]
- 4. 素数を最初から 100 番目まで求めて表示する MIPS のアセンブリ言語プログラムを作成してテストせよ. その際. 素数を求めるためにに表 1 に示す 2 つのルーチンを作成すること.
- 5. 素数を最初から 100 番目まで求めて表示する MIPS のアセンブリ言語プログラムを作成してテストせよ. ただし, 配列に実行結果を保存するように main 部分を改造し, ユーザの入力によって任意の番目の配列 要素を表示可能にせよ.

2 プログラムの説明

使用した MIPS アセンブリ言語のソースコードは、6章に示す.

表1 課題4で実装する2つのルーチン

関数名	概要
test_prime(n)	n が素数なら 1, そうでなければ 0 を返す
main()	整数を順々に素数判定し,100 個プリント

2.1 課題 1-1

まず、A.8節 「入力と出力」に示されている方法で実装したコードについて説明する [1]. このプログラムは起 動すると, msg ラベルに対応するメモリ上のアドレスを 1a 命令により, \$a0 レジスタにロードする. 次に move 命令を用いて, main を呼び出したコンソールに戻るためのアドレスを\$s0 レジスタに退避する. 続いて, loop ラ ベルから始まるループ部に突入する. ここでは 1b 命令を用いて、\$a0 レジスタが指すメモリ上のアドレスに記録 されているバイトを\$a1 レジスタにロードする. ロードしたときの値が0であった場合, begz 命令での分岐条件 を満たし, end ラベルに対応するメモリ上のアドレスの命令にジャンプする. 条件を満たさない場合は, jal 命令 で print ラベルにジャンプするとともに、\$ra レジスタに jal print の次の命令が存在するメモリ上のアドレス を代入する. print ラベルから始まる表示の処理では、まず lw 命令を用いて、メモリ上のアドレス 0xffff0008 に存在するワードを\$t0 レジスタにロードする. このとき, \$t0 レジスタに入っている値は1かそれ以外であり, これが1であればプリンタは動作可能な状態である. li 命令で,即値1を\$t1レジスタにロードして, and 命令 で\$t0 レジスタと\$t1 レジスタの論理積をとって\$t0 レジスタに代入する. 次の begz 命令で\$t0 レジスタの値 が 0 である場合、プリンタの準備はできていないものとし、もう一度 print ラベルに対応するメモリ上のアドレ スの命令に戻って準備を試みる. 準備ができてい場合は、sw 命令でメモリ上のアドレス 0xffff000c に表示した い 1 文字の ASCII コードを送る. 送った瞬間に SPIM のコンソールに文字が表示される. そして, j \$ra によ り呼び出し元である main 中の jal print の次の命令にジャンプする. ここまでで1文字表示できたので,次の 文字情報を得るために addi 命令で\$a0 レジスタの中身に 1 を加え、次に 1b 命令でロードする ASCII コードの 参照アドレスを変化させる. そして j 命令で loop のラベルにジャンプする. ループを抜けた end ラベル部では, \$s0 に退避していたコンソールに戻るためのアドレスを\$ra から復元し、j 命令でコンソールに戻る.

続いて、A.9 節 最後「システムコール」に示されている方法で実装したコードについて説明する [1]. このプログラムは起動すると、msg ラベルに対応するメモリ上のアドレスを 1a 命令により、\$a0 レジスタにロードする. 次に syscall 命令を実行したときに参照されるシステム・コール・コードを\$v0 レジスタに予めロードしておく、今回のシステム・コール・コードは 4 で、 $print_string$ サービスを指定する. syscall 命令でシステムコールを実行し、文字列が出力される. 文字列を出力できるため、アドレスを 1 文字ずつずらすような処理は行わない. j 命令でコンソールに戻る.

2.2 課題 1-2

2.3 課題 1-3

まず、move 命令を使って、\$ra レジスタの中身を\$s0 レジスタに退避する。次に 1a 命令で msg のラベルに対応するメモリ上のアドレスを 1a0 レジスタにロードする。1gal 命令を使ってこの命令の次の命令に対応するメモリ上のアドレスを1a2 レジスタに保存し、ラベル 1a3 print_str に対応するメモリ上のアドレスの命令にジャンプする。1a4 を代入して、1a4 を代入して、1a5 などない。1a6 で1a6 で1a7 で1a7 で1a7 で1a7 で1a7 で1a7 で14 で 14 で 1

を代入する. 再び jal 命令で、この命令の次の命令に対応するメモリ上のアドレスを\$ra レジスタに保存し、ラ ベル fact に対応するメモリ上のアドレスの命令にジャンプする. ここで fact ラベルから始まる処理は、再起呼 び出しを受ける可能性のある処理であるから自身を呼び出したときの\$ra レジスタの値や、再起呼び出しから戻 る際に必要となる数値の情報はスタックに保存しておく必要がある. subu \$sp, \$sp, 32 から始まる 5 行の処 理はスタックに関する命令である. まず, subu 命令でスタックポインタが指すメモリ上のアドレスを 32 バイト 下げる. これによりスタックが 32 バイト確保された. 続いて, sw 命令でメモリ上に対応するレジスタの値をコ ピーする. ここではスタックポインタを基準として 20 番地上の(つまりオフセットが 20) アドレスに\$ra レジ スタのコピー, 16番地上のアドレスに\$fp レジスタのコピー, 24番地上のアドレスに\$a0 レジスタのコピーを 取っている. なお、この\$a0 レジスタは C 言語における fact() 関数の引数部に相当する値である. 最後に addu命令でフレームポインタを更新している. ここからは fact ラベル部の内部処理に移る. まず, li 命令で\$t0 レ ジスタに1を代入する. これは後の分岐命令で用いる値である. bgeu 命令で\$a0 レジスタの値が\$t0 レジスタよ り小さいとき, すなわち\$a0 の値が 1 より小さいとき, else のラベルに対応するメモリ上のアドレスの命令に ジャンプする. そうでない場合は li 命令で\$v0 レジスタに l を代入する. その後, lw 命令でスタックの内容を 復元し、スタックを開放し、j \$raで呼び出し元に戻る. else ラベルにジャンプした場合は、addi命令で\$a0 レジスタに -1 を加えて(すなわち 1 を引いて),jal 命令で fact ラベルにジャンプする.再起呼び出しが終了 し、戻ってくると 40 行目から下の 1w 命令からスタックの復元と解放の処理が始まる. \$ra レジスタ, \$fp レジ スタ, \$a0 レジスタに対応する値を復元する. そして, fact の計算のために, mulo 命令で\$v0 レジスタと\$a0 レジスタを乗算する. そして j \$ra で復元した\$ra レジスタの情報を元に呼び出し元に戻る. 再起呼び出しから 戻る過程で、\$v0 レジスタは自身ともう一つのレジスタの値でしか更新されないようになっている. 再起呼び出 しから完全に戻り、ラベル main まで戻ると、move 命令で\$v0 レジスタの値を\$a0 レジスタにコピーする.次に 実行する syscall 命令の引数で\$v0 レジスタの値は上書きされるため、別のレジスタにコピーするとともに実際 に表示するときに使う\$a0 レジスタにコピーしている. そして, jal 命令を使って, print_int ラベルに対応す るメモリ上のアドレスの命令にジャンプする.このとき, \$ra レジスタにこの次の命令のアドレスを格納する. print_int のラベルからの処理では、li 命令で\$v0 レジスタに1を代入して、syscall 命令でシステムコールを 実行する. j \$raで, print_intを呼び出したその一つ次の命令に戻る. 最後に改行するために la命令で msg2 のアドレスをロードする. そして jal print_str で print_str ラベルに対応する命令にジャンプし, syscall 命令を実行する. 最後に move 命令で\$s0 レジスタからコンソールに戻るためのアドレスを\$ra から復元し, j 命 令でコンソールに戻る.

2.4 課題 1-4

まず、move 命令を使って、\$ra レジスタの中身を\$s2 レジスタに退避する. 続いて、1i 命令で\$s0 レジスタに 2、\$s1 レジスタに 1 を代入する. このときの\$s0 レジスタの値は検索する数、\$s1 レジスタの値は素数の個数カウントに使用する. move 命令を使って、\$s0 レジスタの値を\$a0 レジスタにコピーする. そして、ja1 test_prime で test_prime ラベルに対応するメモリ上のアドレスの命令にジャンプするとともに、この次の命令のアドレスを\$ra レジスタに格納する. test_prime ラベルから始まる処理では、まず 1i 命令で即値 2 を\$t0 レジスタに ロードする. そして、move 命令で\$a0 レジスタの値を\$t1 レジスタにコピーする. ここから先は prime_loop ラベルの後の命令となり、ループされる処理である. まず、rem 命令で\$t1 レジスタの値を\$t0 レジスタの値で割って、その余りを\$t2 レジスタに代入する. 続いて分岐命令 beq で、\$t0 レジスタと\$t1 レジスタの値が等しいとき、すなわち素数であるとき、j 命令で prime_r1 ラベルに対応するメモリ上のアドレスの命令にジャンプする. その次の beqz 命令は、\$t2 レジスタが 0 のとき、すなわち割り切れて素数ではないとき、prime_r2 ラベルに対応するメモリ上のアドレスの命令にジャンプする. いずれの条件にも合致しない場合は addi 命令で\$t0 レジスタに 1 を加えて、j prime_loop でループの初めにジャンプする. prime_r1 ラベルから始まる処理は、1i 命令で\$v0 レジスタに即値 1 をロードし、prime_end ラベルにジャンプする. prime_r2 ラベルから始まる処理は、1i 命令で\$v0 レジスタに即値 0 をロードする. prime_end の処理は、j \$ra により呼び出し元に戻る処理は、1i 命令で\$v0 レジスタに即値 0 をロードする. prime_end の処理は、j \$ra により呼び出し元に戻る処理

である。main のループ部に処理が戻ってくると、beqz 命令が実行される。\$v0 レジスタの値が0 ならば、r1 ラベルに対応するメモリ上のアドレスの命令にジャンプする。r1 ラベル内の処理では、bge 命令で\$s1 レジスタが 100 以上のときに、end ラベルに対応するメモリ上のアドレスの命令にジャンプする。そうでなければ addi 命令で\$s0 レジスタに 1 を加え、j 1 oop でループの初めにジャンプする。r1 に分岐しなかった場合は、数値が素数 だったときで、addi 命令で\$s1 レジスタに 1 を加える。そして、move 命令で\$s0 レジスタの値を\$a0 レジスタにコピーする。 $print_int$ ラベルと $print_str$ ラベルに対応した処理が行われる。このとき、 $print_str$ で表示する文字列は space ラベルに対応するメモリ上のアドレスの文字列である。 $print_int$ ラベルと $print_str$ ラベルの処理は、 $print_str$ $print_str$ prin

2.5 課題 1-5

主に 2.4 節とは異なる処理をしている部分に着目して説明する.

7 行目で la 命令を使い、配列の先頭部分を指すメモリ上のアドレスを\$s3 レジスタにロードする. 12 行目から 始まる配列へのデータ格納処理は、まず move 命令で\$s1 レジスタの値を\$t0 レジスタにコピーする.次に mulo 命令でこの値を 4 倍する. ここまででオフセットの計算を行う. そして add 命令で\$t0 レジスタと s3 レジス タの値を足して、配列の添字に対応するメモリ上のアドレスに変換する.この値を\$t0 レジスタに代入する. そ して sw 命令で\$t0 の値に対応するメモリ上のアドレスに\$s0 の値をコピーする. 新たに追加された loop2 ラベ ルから始まる処理ではユーザからの入力を受け付ける.まず入力を受け付けていることを>により示す.これを 表示するために 1a 命令で indicate ラベルに対応するメモリ上のアドレスを\$a0 レジスタにロードする.次に print_str に対応する処理を行う. そして, jal 命令で read_int のアドレスにジャンプする. read_int ラベ ルに対応する処理では、li命令で5を代入しsyscall命令でシステムコールを実行する.このとき、入力され た整数値は\$v0 レジスタに代入される. j \$ra で呼び出し元である loop2 ラベル内の処理に戻る. bge 命令と ble 命令で例外用の分岐処理を行う. bge 命令は\$v0 レジスタが 101 以上のとき, ble 命令は\$v0 レジスタが 0 以下のとき, end ラベルにジャンプする. そうでなければ\$v0 レジスタの整数値を元に対応するメモリ上のアド レスを計算する.まず、\$v0 レジスタの値を\$t0 レジスタにコピーする.そして、\$t0 レジスタから1を引く. 次に mulo 命令でこの値を 4 倍する.そして mulo 命令でmulo かジスタの値を足して,配列の添字に 対応するメモリ上のアドレスに変換する. そして\$t0 レジスタの情報を元にメモリ上の値にアクセスし、1w 命令 で\$a0 レジスタに値をロードする. そして print_int に対応する処理を行う. 続いて la 命令で line に対応す るメモリ上のアドレスを\$a0 レジスタにロードし, print_str に対応する処理を行う. そして j loop2 で loop2 の処理の先頭にジャンプする. 素数であれば 1 を返す処理である 45-60 行の部分は, 6.4 節に示す課題 1-4 のコー ドの26-41 行目と全く同じ処理をするため、ここでは説明しない. end ラベルに対応する処理では、最 move 命令 で\$s2 レジスタからコンソールに戻るためのアドレスを\$ra から復元し、j 命令でコンソールに戻る.

3 プログラムの使用法と実行結果

プログラムは、CentOS 7.6.1810 (Core) の xspim で動作を確認している. まず、ターミナルに xspim -mapped_io&と打ち込んで、xspim を実行する. 実行後に load の機能を使い、拡張子が.s のアセンブリファイルを読み込む. run の機能で読み込んだプログラムを走らせる. プログラムを走らせた後、もう一度プログラムを走らせる場合には clear でメモリとレジスタの値を初期化した後、再度ロードする必要がある.

3.1 課題 1-1

A.8 節 「入力と出力」に示されている方法と、A.9 節 最後「システムコール」に示されている方法のいずれの方法でも SPIM のコンソールの出力は Hello World となることを確認した.

3.2 課題 1-2

6.2 節に示したコードを実行すると、コンソール出力は 12 となる.それに対して、9 行目の.data がない場合は、コンソール出力は 1212 となった.表示した後、プログラムは終了する.

3.3 課題 1-3

実行すると, コンソール出力は The factorial of 10 is 3628800 となる. 表示した後, プログラムは終了する.

3.4 課題 1-4

実行すると, 2 から始まる合計 100 個の素数をコンソールに表示し,終了する. コンソールに表示される素数 は次のとおりであり,適宜改行している. 実際は 541 の表示後に一度だけ改行する. 表示した後,プログラムは終了する.

2 3 5 7 11 13 17 19 23 29 31 37 41 43 47 53 59 61 67 71 73 79 83 89 97 101 103 107 109 113 127 131 137 139 149 151 157 163 167 173 179 181 191 193 197 199 211 223 227 229 233 239 241 251 257 263 269 271 277 281 283 293 307 311 313 317 331 337 347 349 353 359 367 373 379 383 389 397 401 409 419 421 431 433 439 443 449 457 461 463 467 479 487 491 499 503 509 521 523 541

3.5 課題 1-5

実行すると、コンソールに>が表示され、入力待ちになる。ここに 1-100 の数値を入力できる。入力後に Enterキーを押すと、対応する素数が 1 つ表示される。0 以下か 101 以上の値を入力するとプログラムが終了する。以下に入力と出力の一例を示す。

>23

83

>81

419

>0

4 考察

4.1 課題 1-1

A.8 節 「入力と出力」に示されている方法に示されている方法で実装する場合,空行を除くコードは 25 行である.それに対して,A.9 節 最後「システムコール」に示されている方法で実装する場合,空行を除くコードはたった 12 行となった [1].これは前者が 1 文字ずつしか表示できないため,1 文字ずつ読み込む処理とアドレス

特に「入力と出力」に示されている方法で実装する場合には、0xffff0008 番地の確認を怠ると、プリンタの故障につながることがあったり、計算機によって異なるアドレスをとる可能性のある 0xffff0008 番地をプログラムのコード内に書いてしまうことになったりするなどの問題もある。また、他のプログラムと同時に走行した場合の優先順序をつけられないために競合が発生することがある。システムコールを用いていれば、表示するタイミングや他のプログラムとの優先順序もオペレーティングシステムが決めてくれるため、プログラムを書く際に深く意識する必要はない。

4.2 課題 1-2

.text は、これに後続する項目をテキスト・セグメントの中に格納する指令である [1]. テキスト・セグメントとは、プログラムの命令を格納する領域のことを指す.

.data は、これに後続する項目をデータ・セグメントの中に格納する指令である [1]. データ・セグメントは、テキスト・セグメント上に位置していて、特に静的データ・セグメントは開始アドレス 0x10000000 から始まる領域に格納される. 今回の.word 12 をデータ・セグメントに配置するようにした場合、このアドレスの領域に格納されることになる.

.align は,これに後続するデータを 2^n バイト境界に整列化する命令である [1]. MIPS R2000 の多くのロード・ストア命令は,整列化されたデータのみを操作の対象としているため,ワードのデータアドレスは 4 の倍数でなければならない.つまり.align 2 をすれば, 2^2 すなわち 4 バイト境界に整列化することになり,ロード・ストア命令が正しく実行できるようになる.また,SPIM シミュレータは MIPS32 アーキテクチャを実現したプロセッサ用に書かれたアセンブリ言語を実行するシミュレータであるため,ワード単位は 32bit すなわち 4 バイトである.このため,どの命令もデータも 4 バイト境界から始まっておくべきである.

6.2 節のコード中の 9 行目の.data がない場合,.word 12 はテキスト・セグメント中に配置されることとなり,プログラムの一部となる.ジャンプ命令を実行しない場合は,プログラムカウンタを 4 バイトずつ増やしながら命令を実行していくため,.word 12 も命令であるかのように実行しようとする.MIPS32 アーキテクチャの命令は,4 バイト長であり,.word 12 もがイト長は 4 バイトでそこに 10 進数の 12 があるだけである.同じ長さであるから命令として解釈することも可能になる.数値 12 を命令として解釈すると p.611 の図 p.611 ののの p.611 ののの p.611 の図 p.611 の図 p.611 ののの p.611 ののの p.611 ののの p.611 のの p.6

5 感想

アセンブリ言語を書くのは難しく感じたが、課題に取り組んでいるうちに使い方やプログラムの動きも分かるようになり、楽しかった、課題 1-2 を考察するのが大変だった、疑似命令は便利な命令だと感じた.

6 作成したプログラムのソースコード

使用したプログラムを以下に添付する.

6.1 課題 1-1 で用いたコード

下記は、「入力と出力」に示されている方法で実装した例.

```
# データセグメント
      data
1 .
                          #4バイト境界に揃える
2:
      .align 2
3: msg:
      .asciiz "Hello World" # 出力文字情報
4:
5:
                          # テキストセグメント
                          #4バイト境界に揃える
      .align 2
7:
8: main:
            $a0, msg
                          # a0 に msg のアドレスを格納する
9:
      la
                          # $s0 に main を呼び出した元のアドレスを格納
            $s0, $ra
10:
      move
11: loop:
            $a1, 0($a0) # a0 の指し先の値を$a1 にロードする
     beqz $a1, end # a1が0のとき, endへ
jal print # printのアドレスにジャンプ(次の命令のアドレスを$raに)
addi $a0, 1 # $a0に1を加える
i loop # mainのアドレスにジャンプ
13:
     jal
14.
            loop
                          # main のアドレスにジャンプ
16:
17: print:
            $t0, 0xfffff0008 # 0xfffff0008 番地にあるワードを$t0 にロードする
18:
     lw
19:
      li
           $t1, 1 # $t1 に 1 を代入する
            $t0, $t0, $t1 # $t0 と$t1 の論理積をとって$t0 に代入する ($t0 が 1 か確認する)
      and
20:
21:
      beqz
             $t0, print # $t0 が 0 のとき, もう一度準備を試みる
22:
      sw
             $a1, 0xfffff000c # 0xfffff000c 番地に$a1 にあるワードを送る
                      # 呼び出し元に戻る
23:
            $ra
24: end:
   move $ra, $s0 # main を呼び出した元のアドレスを$ra に復元
                          # コンソールに戻る
26:
     j
           $ra
```

下記は、「システムコール」に示されている方法で実装した例.

```
# データセグメント
      .data
                        #4バイト境界に揃える
2:
     .align 2
3: msg:
4:
     .asciiz "Hello World" # 出力文字情報
     .text
                        # テキストセグメント
6:
                        #4バイト境界に揃える
7:
8: main:
                     # msg のアドレスを $a0 に格納
# print_stringを指定
           $a0, msg
9:
   la
    li
          $v0, 4
                      # システムコールの実行
11:
    syscall
                       # コンソールに戻る
    j $ra
12:
```

6.2 課題 1-2 で用いたコード

```
# テキストセグメント
      text
1 •
                        #4バイト境界に揃える
2:
      .align 2
4: _print_data:
5:
    la $a0, ghost
                       # ghost のアドレスを$a0 にロード
            $a0, 0($a0)
                        # $a0 レジスタに対応するメモリ上のアドレスのワードを自身にロード
           $v0, 1
7:
     li
                        # $v0 レジスタに 1 を代入
8:
                        # データセグメント
10:
     .align 2
                        #4バイト境界に揃える
11: ghost:
```

```
12.
      word 12
                        # 12 (1 ワード)
13:
                        # テキストセグメント
14:
     .text
                        # システムコールの実行
15:
      syscall
      j $ra
                        # 呼び出し元に戻る
16:
                        # スタート
18: main:
            $sp, $sp, 24
                        # スタックを積む
19.
    subu
                        # $sp + 16 番地のアドレスに$ra の値をバックアップ
     sw
           $ra, 16($sp)
     jal
            _print_data
                        # _print_data のアドレスにジャンプ(次の命令のアドレスを$raに)
     lw
            $ra, 16($sp)
                       # $sp + 16 番地のアドレスから$ra の値を復元
22:
                      # スタックを解放
23:
     addu
            $sp, $sp, 24
            $ra
                        # 呼び出し元に戻る
24:
      j
```

6.3 課題 1-3 で用いたコード

```
# テキストセグメント
1:
                          #4バイト境界に揃える
2:
      .align 2
3: main:
            $s0, $ra
                          # main を呼んだ戻り先のアドレスを$s0 に保存しておく
4:
    move
5:
                          # msg のアドレスを$a0 にロード
     la
           $a0, msg
6:
                          # print_str のアドレスにジャンプ (次の命令のアドレスを$raに)
7:
     jal
            print_str
8:
                          # $a0 に 10 を代入(任意の int 値)
9:
      li
            $a0, 10
10:
11:
      jal
            fact
                          # fact に飛ぶ
             $a0, $v0
                          # $a0 に$v0 の値をコピー
12:
      move
13:
      jal
            print_int
                          # print_int のアドレスにジャンプ (次の命令のアドレスを$raに)
14:
                          # msg2 のアドレスを$a0 にロード
15:
      la
            $a0, msg2
     jal
            print_str
                          # print_str のアドレスにジャンプ (次の命令のアドレスを$raに)
16:
17:
                          # main を呼んだ戻り先のアドレスを$ra に代入
18:
      move
             $ra, $s0
                          # コンソールに戻る
19.
      j
             $ra
20:
21: fact:
                          # 32 バイト確保
             $sp, $sp, 32
22.
     subu
23:
      SW
             $ra, 20($sp)
                          # $sp + 20 のアドレスにもともとの呼出しアドレスを保存
      sw
             $fp, 16($sp)
                          # $sp + 16 のアドレスに$fp を保存
24:
             $a0, 24($sp)
                          # $sp + 24 のアドレスに$a0 を保存
25:
      SW
      addu
                          # 新しく$fp を設定
            $fp, $sp, 28
26:
            $t0, 1
                          # $t0 に 1 を代入 (分岐命令用)
28:
      li
29.
             $a0, $t0, else # $a0 が$t0 より小さいとき, else へ
30:
      bgeu
31:
      li
             $v0, 1
                          # $v0 に 1 を代入
             $ra, 20($sp)
                         # 保存しておいた$sp を復元
      ٦w
32.
33:
     lw
             $fp, 16($sp) # 保存しておいた$fp を復元
34:
     addu
             $sp, $sp, 32 # スタックを解放
                          # 呼び出し元に戻る
35:
     j
            $ra
36: else:
             $a0, -1
      addi
                          # $a0 から1を引く
37:
                          # 再帰呼び出しをする
38.
      jal
             fact
39:
      lw
             $ra, 20($sp)
                          # 保存しておいた$sp を復元
41:
      ٦w
             $fp, 16($sp)
                         # 保存しておいた$fp を復元
      lw
                         # 対応する$a0 の値を復元
42:
             $a0, 24($sp)
                          # スタックを解放
43:
      addu
             $sp, $sp, 32
44:
45.
      mulo
            $v0, $a0, $v0 # $v0 と$a0 をかけて$v0 に代入
46:
      j
             $ra
                          # 呼び出し元に戻る
47:
48: print_int:
49: li
             $v0, 1
                         # $v0 に 1 を代入
                          # システムコールの実行
50:
      syscall
                          # 呼び出し元に戻る
51:
      j
             $ra
```

```
52: print_str:
                        # $v0 に 4 を代入
53:
     li
             $v0, 4
                           # システムコールの実行
      syscall
                           # 呼び出し元に戻る
55:
             $ra
56:
                           # データセグメント
57:
      .data
58:
      .align 2
                           #4バイト境界に揃える
59: msg:
      .asciiz "The factorial of 10 is "
60:
                           #4バイト境界に揃える
62:
      .align 2
63: msg2:
      .asciiz "\n"
64:
```

6.4 課題 1-4 で用いたコード

```
# テキストセグメント
1:
       .text
                            #4バイト境界に揃える
2:
       .align 2
3: main:
                           # $s2 に main を呼び出した元のアドレスを格納
4:
      move
              $s2, $ra
                           # $s0 に 2 を格納(検索する数)
5:
       li
              $s0, 2
                           # $s1 に 0 を格納 (素数の個数カウント)
              $s1, 0
6:
      lί
7: loop:
8:
      move
             $a0, $s0
                           # $s0 の値を$a0 にコピー
                        # test_prime のアドレスにジャンプ(次の命令のアドレスを$ra に)
# $v0 が 0 ならば,r1 に分岐
9:
       jal
             test_prime
10:
       beqz
              $v0, r1
11:
       addi
             $s1, 1
                           # $s1 に 1 を加える
             $a0, $s0
                           # $s0 の値を$a0 にコピー
12:
      move
      jal print_in_ $a0, space
13:
                           # print_int のアドレスにジャンプ(次の命令のアドレスを$ra に)
14:
                        # space のアドレスを$a0 にロード
      jal print_str
                           # print_str のアドレスにジャンプ(次の命令のアドレスを$ra に)
15:
16: r1:
17:
      bge
             $s1, 100, end # $s1 が 100 以上のとき, end にジャンプ
              $s0, 1
                            # $s0 に 1 を加える
18:
       addi
                            # loop のアドレスにジャンプ
19:
       j
              loop
20: end:
             $a0, line # line のアドレスを$a0 にロード
print_str # print_str のアドレスにジャンプ(次の命令のアドレスを$ra に)
$ra, $s2 # main を呼び出した元のアドレスを$ra に復元
            $a0, line
21:
22.
      jal
      move
23:
                           # コンソールに戻る
24:
       j
              $ra
25:
26: test_prime:
              $t0, 2
                           # $t0 に 2 を代入
27: li
                            # $t1 に$a0 の値をコピー
28:
      move
              $t1, $a0
29: prime_loop:
              $t2, $t1, $t0 # $t1 を$t0 で割った余りを$t2 に代入する
30:
      rem
              $t0, $t1, prime_r1 # $t0 と$t1 が等しいとき prime_r1 のアドレスにジャンプ
31:
       beq
              $t2, prime_r2 # $t2が0ならprime_r2のアドレスにジャンプ
32.
      begz
                           # $t0 に 1 を加える
33:
      addi
              $t0, 1
              prime_loop
                           # prime_loop のアドレスにジャンプ
34:
      j
35: prime_r1:
   li
36:
              $v0, 1
                           # $v0 に 1 を代入
                            # prime_end のアドレスにジャンプ
37:
       j
              prime_end
38: prime_r2:
              $v0, 0
                            # $v0 に 0 を代入
39: li
40: prime_end:
                           # 呼び出し元に戻る
41:
     j
              $ra
42:
43: print_int:
              $v0, 1
44:
      li
                           # $v0 に 1 を代入
                            # システムコールの実行
45:
       syscall
      j
                           # 呼び出し元に戻る
46:
              $ra
47: print_str:
                         # $v0 に 4 を代入
              $v0, 4
48:
     li
49:
      syscall
                           # システムコールの実行
                           # 呼び出し元に戻る
50:
       j
              $ra
51:
```

```
52: .data # データセグメント
53: .align 2 # 4バイト境界に揃える
54: space:
55: .asciiz " "
56: .align 2 # 4バイト境界に揃える
57: line:
58: .asciiz "\n"
```

6.5 課題 1-5 で用いたコード

```
# テキストセグメント
                          #4バイト境界に揃える
2:
      .align 2
3: main:
             $s2, $ra
                          # $s2 に main を呼び出した元のアドレスを格納
4:
      move
                          # $s0 に 2 を格納 (検索する数)
             $s0, 2
5:
      li
                          # $s1 に 0 を格納(素数の個数カウント)
             $s1, 0
6.
      ٦i
             $s3, prime_array# $s3 に prime_array のアドレスをロード
7:
      la
8: loop:
9:
                          # $s0 の値を$a0 にコピー
      move
             $a0, $s0
                          # test_prime のアドレスにジャンプ (次の命令のアドレスを$raに)
10:
      ial
             test_prime
11:
      beqz
             $v0, r1
                          # $v0 が 0 ならば, r1 に分岐
      # ここで配列にデータを格納
12:
13:
      move
             $t0, $s1
                       # $s1 の値を$t0 にコピー
14:
      mulo
             $t0, $t0, 4
                          # $t0 を 4 倍(4byte 区切り)
             $t0, $t0, $s3 # $t0 と$s3 を足して$t0 に格納
15:
      add
             $s0, 0($t0)
16:
      SW
                          # $s0 の値を配列上にコピー
17:
      addi
             $s1, 1
                          # $s1 に 1 を加える
18: r1:
             $s1, 100, 10op2 # $s1 が 100 より大きいとき, 10op2 にジャンプ
19:
      bgt
             $s0, 1 # $s0 に 1 を加える
20:
      addi
             loop
                          # loop のアドレスにジャンプ
21:
22: loop2:
23:
      la
             $a0, indicate # indicate のアドレスを$a0 にロード
                          # print_str のアドレスにジャンプ(次の命令のアドレスを$raに)
24:
      jal
             print_str
      #入力受付
25:
                          # read_int のアドレスにジャンプ (次の命令のアドレスを$raに)
26:
      ial
            read int
      # 例外確認
27:
            $v0, 101, end # 入力が 101 以上の場合,終了
28.
      bge
             $v0, 0, end
                          # 入力が ○以下の場合,終了
29:
      ble
      # 正しい番地を計算
30:
                          # $v0 の値を$t0 にコピー
31:
      move
            $t0, $v0
             $t0, -1
                          # $t0 から 1 を引く
      addi
32:
             $t0, $t0, 4
                         # $t0 を 4 倍(4byte 区切り)
33:
      mulo
             $t0, $t0, $s3 # $t0 と$s3 を足して$t0 に格納
34:
      add
      # 配列から読みだした値を表示
35:
           $a0, 0($t0) # $a0 に対応する配列上の値をロード
36:
      lw
37:
      jal
             print_int
                          # print_int のアドレスにジャンプ(次の命令のアドレスを$ra に)
            $a0, line
                         # line のアドレスを$a0 にロード
38:
      la
                          # print_str のアドレスにジャンプ(次の命令のアドレスを$ra に)
      jal
          print_str
39:
            loop2
40:
      j
41: end:
42:
                          # main を呼び出した元のアドレスを$ra に復元
   move
             $ra, $s2
                          # コンソールに戻る
43:
      j
             $ra
44:
45: test_prime:
             $t0, 2
                          # $t0 に 2 を代入
46:
     li
47:
      move
             $t1, $a0
                          # $t1 に$a0 の値をコピー
48: prime_loop:
             $t2, $t1, $t0 # $t1 を$t0 で割った余りを$t2 に代入する
49:
      rem
             $t0, $t1, prime_r1 # $t0 と$t1 が等しいとき prime_r1 のアドレスにジャンプ
50:
      beq
             $t2, prime_r2 # $t2が0ならprime_r2のアドレスにジャンプ
51:
      beqz
52:
      addi
             $t0, 1
                          # $t0 に 1 を加える
      j
             prime_loop
                         # prime_loop のアドレスにジャンプ
54: prime_r1:
   li
55:
             $v0, 1
                          # $v0 に 1 を代入
             , -
prime_end
                          # prime_end のアドレスにジャンプ
56:
      j
57: prime_r2:
```

```
58: li
           $v0,0 #$v0に0を代入
59: prime_end:
60:
                     # 呼び出し元に戻る
     j
61:
62: print_int:
   li
           $v0, 1
                      # $v0 に 1 を代入
63:
                      # システムコールの実行
64:
     syscall
           $ra
                      # 呼び出し元に戻る
65:
     j
66: print_str:
67: li
           $v0, 4
                     # $v0 に 4 を代入
                      # システムコールの実行
    syscall
68:
                      # 呼び出し元に戻る
69:
    j
           $ra
70: read_int:
71: li
72: syscall
           $v0,5 # read_int (戻り値は$v0に,1-100のみ入力可) # 読み取り
     syscall
72:
    j $ra
73:
                      # 呼び出し元に戻る
74:
75: .data
                      # データセグメント
                      # 4 バイト境界に揃える
76:
     .align 2
77: indicate:
78: .asciiz ">"
                     #4バイト境界に揃える
79:
     .align 2
80: line:
81: .asciiz "\n"
                      # 4 バイト境界に揃える
82:
     .align 2
83: prime_array:
84: .space 400
```

参考文献

[1] David A. Patterson, John L. Hennessy, コンピュータの構成と設計 第5版[下] -ハードウエアとソフトウエア-, 日経 BP 社, 2014.